query: #Units used by property P2405
  #by Jura1, 2016-01-24, rev. 2020-10-05, 
  
  SELECT DISTINCT ?unit ?unitLabel ?count ?has_SI_conversion ?has_some_conversion ?allowed_unit ?sample ?sampleLabel
  WHERE
  {
    {
        SELECT ?unit (count(DISTINCT(?item)) as ?count) (SAMPLE(?item) as ?samplest)
      WHERE
      {
            ?item psv:P2405/wikibase:quantityUnit ?unit
      }
      GROUP BY ?unit
    }
      OPTIONAL { ?unit wdt:P2370 [] . BIND("y" as ?has_SI_conversion) }
      OPTIONAL { ?unit wdt:P2442 [] . BIND("y" as ?has_some_conversion) }
      ?sample p:P2405 ?samplest .
      OPTIONAL { wd:P2405 p:P2302 [ ps:P2302 wd:Q21514353 ; pq:P2305 ?unit ] . BIND( "y" as ?allowed_unit) }
      SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en"}
  }
  ORDER BY DESC(?count) ?unit
  LIMIT 10